// File: /home/nicklatkovich/pixelplex/riochain/om-token-v2/contracts/IntervalEscrow.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.6.12;
pragma experimental ABIEncoderV2;

import "openzeppelin-solidity/contracts/math/Math.sol";
import "openzeppelin-solidity/contracts/math/SafeMath.sol";
import "openzeppelin-solidity/contracts/token/ERC20/IERC20.sol";
import "openzeppelin-solidity/contracts/token/ERC20/SafeERC20.sol";
import "./TwoStageOwnable.sol";

contract IntervalEscrow is TwoStageOwnable {
    using SafeMath for uint256;
    using SafeERC20 for IERC20;

    struct NominatedStrategy {
        bool nominated;
        uint256 applicableAt;
        uint256 firstIntervalPerBlockReleaseAmount;
        uint256[] intervals;
    }

    struct Strategy {
        uint256 currentIntervalIndex;
        uint256 currentIntervalStartedAt;
        uint256 lastReleasedAt;
        uint256 perBlockReleaseAmount;
        uint256[] intervals;
    }

    function getBlockNumber() internal virtual view returns (uint256) {
        return block.number;
    }

    function getTimestamp() internal virtual view returns (uint256) {
        return block.timestamp;
    }

    uint256 public exitApplicableAt = 0;
    uint256 public exitTimeout;
    uint256 public strategyChangeTimeout;
    NominatedStrategy public nominatedStrategy;
    Strategy public currentStrategy;

    uint256 private _pool = 0;
    uint256 private _released = 0;
    IERC20 private _token;

    function pool() public view returns (uint256) {
        return _pool;
    }

    function released() public view returns (uint256) {
        return _released;
    }

    function token() public view returns (IERC20) {
        return _token;
    }

    function calculateUpdates(uint256 limit)
        public
        view
        returns (Strategy memory updatedStrategy, uint256 releasedAmount)
    {
        uint256 currentBlockNumber = getBlockNumber();
        updatedStrategy = currentStrategy;
        uint256 intervalsCount = updatedStrategy.intervals.length;
        while (limit > 0 && updatedStrategy.lastReleasedAt < currentBlockNumber) {
            uint256 releaseBlockNumber;
            uint256 perBlockReleaseAmount = updatedStrategy.perBlockReleaseAmount;
            if (updatedStrategy.currentIntervalIndex >= intervalsCount) releaseBlockNumber = currentBlockNumber;
            else {
                uint256 currentIntervalLength = updatedStrategy.intervals[updatedStrategy.currentIntervalIndex];
                uint256 currentIntervalStartedAt = updatedStrategy.currentIntervalStartedAt;
                uint256 currentIntervalEndsAt = currentIntervalStartedAt.add(currentIntervalLength);
                if (currentBlockNumber >= currentIntervalEndsAt) {
                    releaseBlockNumber = currentIntervalEndsAt;
                    updatedStrategy.currentIntervalIndex = updatedStrategy.currentIntervalIndex.add(1);
                    updatedStrategy.currentIntervalStartedAt = currentIntervalEndsAt;
                    updatedStrategy.perBlockReleaseAmount /= 2;
                } else releaseBlockNumber = currentBlockNumber;
            }
            uint256 blockDiff = releaseBlockNumber.sub(updatedStrategy.lastReleasedAt);
            uint256 intervalReleasedAmount = blockDiff.mul(perBlockReleaseAmount);
            releasedAmount = releasedAmount.add(intervalReleasedAmount);
            updatedStrategy.lastReleasedAt = releaseBlockNumber;
            limit -= 1;
        }
        releasedAmount = Math.min(releasedAmount, _pool);
    }

    function getReleasableAmountWithLimit(uint256 limit) public view returns (uint256) {
        (, uint256 result) = calculateUpdates(limit);
        return result;
    }

    function getReleasableAmount() public view returns (uint256) {
        return getReleasableAmountWithLimit(uint256(-1));
    }

    event Claimed(uint256 amount);
    event ExitInitialized(uint256 applicableAt);
    event Exited(uint256 amount);
    event PoolIncreased(address indexed payer, uint256 amount);
    event Released(uint256 amount);
    event StrategyNominated(uint256 firstIntervalPerBlockReleaseAmount, uint256[] intervals, uint256 applicableAt);
    event StrategyUpdated(uint256 firstIntervalPerBlockReleaseAmount, uint256[] intervals);

    constructor(
        uint256 exitTimeout_,
        uint256 firstIntervalPerBlockReleaseAmount,
        uint256 strategyChangeTimeout_,
        address owner_,
        uint256[] memory intervals,
        IERC20 token_
    ) public TwoStageOwnable(owner_) {
        exitTimeout = exitTimeout_;
        strategyChangeTimeout = strategyChangeTimeout_;
        _token = token_;
        _setup(firstIntervalPerBlockReleaseAmount, intervals);
    }

    function releaseWithLimit(uint256 limit) public returns (bool success, uint256 releasedAmount) {
        (currentStrategy, releasedAmount) = calculateUpdates(limit);
        _pool = _pool.sub(releasedAmount);
        _released = _released.add(releasedAmount);
        emit Released(releasedAmount);
        success = true;
    }

    function release() public returns (bool success, uint256 _releasedAmount) {
        return releaseWithLimit(uint256(-1));
    }

    function applyExit() public onlyOwner returns (bool success) {
        require(exitApplicableAt > 0, "Exit not initialized");
        require(getTimestamp() >= exitApplicableAt, "Exit timeout not passed");
        exitApplicableAt = 0;
        _released = _released.add(_pool);
        _pool = 0;
        emit Exited(_released);
        _claim(_released);
        return true;
    }

    function applyNominatedStrategy() public onlyOwner returns (bool success) {
        require(nominatedStrategy.nominated, "No nominated strategy");
        require(getTimestamp() >= nominatedStrategy.applicableAt, "Nominating timeout not passed");
        release();
        _setup(nominatedStrategy.firstIntervalPerBlockReleaseAmount, nominatedStrategy.intervals);
        nominatedStrategy.nominated = false;
        return true;
    }

    function exit() public onlyOwner returns (bool success) {
        require(exitApplicableAt == 0, "Exit initilized");
        exitApplicableAt = getTimestamp().add(exitTimeout);
        emit ExitInitialized(exitApplicableAt);
        return true;
    }

    function increasePool(uint256 amount) external returns (bool success) {
        release();
        _increasePool(msg.sender, amount);
        return true;
    }

    function nominateNewStrategy(uint256 firstIntervalPerBlockReleaseAmount, uint256[] memory intervals)
        external
        onlyOwner
        returns (uint256 applicableAt)
    {
        applicableAt = getTimestamp().add(strategyChangeTimeout);
        nominatedStrategy = NominatedStrategy({
            nominated: true,
            applicableAt: applicableAt,
            firstIntervalPerBlockReleaseAmount: firstIntervalPerBlockReleaseAmount,
            intervals: intervals
        });
        emit StrategyNominated(firstIntervalPerBlockReleaseAmount, intervals, applicableAt);
    }

    function claim(uint256 amount) external onlyOwner returns (bool success) {
        release();
        _claim(amount);
        return true;
    }

    function claimAll() external onlyOwner returns (bool success) {
        release();
        _claim(_released);
        return true;
    }

    function _claim(uint256 amount) internal {
        require(_released >= amount, "No enough released tokens");
        _released -= amount;
        emit Claimed(amount);
        _token.safeTransfer(owner, amount);
    }

    function _increasePool(address payer, uint256 amount) internal {
        _pool = _pool.add(amount);
        emit PoolIncreased(payer, amount);
        _token.safeTransferFrom(payer, address(this), amount);
    }

    function _setup(uint256 firstIntervalPerBlockReleaseAmount, uint256[] memory intervals) internal {
        uint256 currentBlockNumber = getBlockNumber();
        currentStrategy = Strategy({
            currentIntervalIndex: 0,
            currentIntervalStartedAt: currentBlockNumber,
            lastReleasedAt: currentBlockNumber,
            perBlockReleaseAmount: firstIntervalPerBlockReleaseAmount,
            intervals: intervals
        });
        emit StrategyUpdated(firstIntervalPerBlockReleaseAmount, intervals);
    }
}

// File: /home/nicklatkovich/pixelplex/riochain/om-token-v2/contracts/OMStakingRewardsEscrow.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.6.12;
pragma experimental ABIEncoderV2;

import "openzeppelin-solidity/contracts/token/ERC20/IERC20.sol";
import "./IntervalEscrow.sol";

contract OMStakingRewardsEscrow is IntervalEscrow {
    constructor(
        uint256 exitTimeout_,
        uint256 firstIntervalPerBlockReleaseAmount,
        uint256 strategyChangeTimeout_,
        address owner_,
        uint256[] memory intervals,
        IERC20 token_
    )
        public
        IntervalEscrow(
            exitTimeout_,
            firstIntervalPerBlockReleaseAmount,
            strategyChangeTimeout_,
            owner_,
            intervals,
            token_
        )
    {}
}

// File: /home/nicklatkovich/pixelplex/riochain/om-token-v2/contracts/TwoStageOwnable.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.6.12;

abstract contract TwoStageOwnable {
  address public nominatedOwner;
  address public owner;

  event OwnerChanged(address indexed newOwner);
  event OwnerNominated(address indexed nominatedOwner);

  constructor(address owner_) internal {
    require(owner_ != address(0), "Owner cannot be zero address");
    _setOwner(owner_);
  }

  function acceptOwnership() external returns (bool success) {
      require(msg.sender == nominatedOwner, "Not nominated to ownership");
      _setOwner(nominatedOwner);
      nominatedOwner = address(0);
      return true;
  }

  function nominateNewOwner(address owner_) external onlyOwner returns (bool success) {
    _nominateNewOwner(owner_);
    return true;
  }

  modifier onlyOwner {
      require(msg.sender == owner, "Ownable: caller is not the owner");
      _;
  }

  function _nominateNewOwner(address owner_) internal {
    nominatedOwner = owner_;
    emit OwnerNominated(owner_);
  }

  function _setOwner(address newOwner) internal {
    owner = newOwner;
    emit OwnerChanged(newOwner);
  }
}

// File: openzeppelin-solidity/contracts/math/Math.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.6.0;

/**
 * @dev Standard math utilities missing in the Solidity language.
 */
library Math {
    /**
     * @dev Returns the largest of two numbers.
     */
    function max(uint256 a, uint256 b) internal pure returns (uint256) {
        return a >= b ? a : b;
    }

    /**
     * @dev Returns the smallest of two numbers.
     */
    function min(uint256 a, uint256 b) internal pure returns (uint256) {
        return a < b ? a : b;
    }

    /**
     * @dev Returns the average of two numbers. The result is rounded towards
     * zero.
     */
    function average(uint256 a, uint256 b) internal pure returns (uint256) {
        // (a + b) / 2 can overflow, so we distribute
        return (a / 2) + (b / 2) + ((a % 2 + b % 2) / 2);
    }
}

// File: openzeppelin-solidity/contracts/math/SafeMath.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.6.0;

/**
 * @dev Wrappers over Solidity's arithmetic operations with added overflow
 * checks.
 *
 * Arithmetic operations in Solidity wrap on overflow. This can easily result
 * in bugs, because programmers usually assume that an overflow raises an
 * error, which is the standard behavior in high level programming languages.
 * `SafeMath` restores this intuition by reverting the transaction when an
 * operation overflows.
 *
 * Using this library instead of the unchecked operations eliminates an entire
 * class of bugs, so it's recommended to use it always.
 */
library SafeMath {
    /**
     * @dev Returns the addition of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `+` operator.
     *
     * Requirements:
     *
     * - Addition cannot overflow.
     */
    function add(uint256 a, uint256 b) internal pure returns (uint256) {
        uint256 c = a + b;
        require(c >= a, "SafeMath: addition overflow");

        return c;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b) internal pure returns (uint256) {
        return sub(a, b, "SafeMath: subtraction overflow");
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting with custom message on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b <= a, errorMessage);
        uint256 c = a - b;

        return c;
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `*` operator.
     *
     * Requirements:
     *
     * - Multiplication cannot overflow.
     */
    function mul(uint256 a, uint256 b) internal pure returns (uint256) {
        // Gas optimization: this is cheaper than requiring 'a' not being zero, but the
        // benefit is lost if 'b' is also tested.
        // See: https://github.com/OpenZeppelin/openzeppelin-contracts/pull/522
        if (a == 0) {
            return 0;
        }

        uint256 c = a * b;
        require(c / a == b, "SafeMath: multiplication overflow");

        return c;
    }

    /**
     * @dev Returns the integer division of two unsigned integers. Reverts on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b) internal pure returns (uint256) {
        return div(a, b, "SafeMath: division by zero");
    }

    /**
     * @dev Returns the integer division of two unsigned integers. Reverts with custom message on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b > 0, errorMessage);
        uint256 c = a / b;
        // assert(a == b * c + a % b); // There is no case in which this doesn't hold

        return c;
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * Reverts when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b) internal pure returns (uint256) {
        return mod(a, b, "SafeMath: modulo by zero");
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * Reverts with custom message when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b != 0, errorMessage);
        return a % b;
    }
}

// File: openzeppelin-solidity/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.6.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `recipient`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `sender` to `recipient` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address sender, address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);
}

// File: openzeppelin-solidity/contracts/token/ERC20/SafeERC20.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.6.0;

import "./IERC20.sol";
import "../../math/SafeMath.sol";
import "../../utils/Address.sol";

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using SafeMath for uint256;
    using Address for address;

    function safeTransfer(IERC20 token, address to, uint256 value) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(IERC20 token, address from, address to, uint256 value) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(IERC20 token, address spender, uint256 value) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        // solhint-disable-next-line max-line-length
        require((value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(IERC20 token, address spender, uint256 value) internal {
        uint256 newAllowance = token.allowance(address(this), spender).add(value);
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(IERC20 token, address spender, uint256 value) internal {
        uint256 newAllowance = token.allowance(address(this), spender).sub(value, "SafeERC20: decreased allowance below zero");
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address.functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        if (returndata.length > 0) { // Return data is optional
            // solhint-disable-next-line max-line-length
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
        }
    }
}

// File: openzeppelin-solidity/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.6.2;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies in extcodesize, which returns 0 for contracts in
        // construction, since the code is only stored at the end of the
        // constructor execution.

        uint256 size;
        // solhint-disable-next-line no-inline-assembly
        assembly { size := extcodesize(account) }
        return size > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        // solhint-disable-next-line avoid-low-level-calls, avoid-call-value
        (bool success, ) = recipient.call{ value: amount }("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain`call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
      return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data, string memory errorMessage) internal returns (bytes memory) {
        return _functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(address target, bytes memory data, uint256 value) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(address target, bytes memory data, uint256 value, string memory errorMessage) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        return _functionCallWithValue(target, data, value, errorMessage);
    }

    function _functionCallWithValue(address target, bytes memory data, uint256 weiValue, string memory errorMessage) private returns (bytes memory) {
        require(isContract(target), "Address: call to non-contract");

        // solhint-disable-next-line avoid-low-level-calls
        (bool success, bytes memory returndata) = target.call{ value: weiValue }(data);
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                // solhint-disable-next-line no-inline-assembly
                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}